```
(* The HolonomicFunctions package can be downloaded from
    http://www.risc.jku.at/research/combinat/software/HolonomicFunctions/ *)
<< RISC`HolonomicFunctions`;
```

HolonomicFunctions Package version 1.7.1 (09-Oct-2013)
written by Christoph Koutschan


```
--> Type  ?HolonomicFunctions  for help.
```

```
(* Define the functions used below. *)
(* Determine the holonomic rank. *)
HolonomicRank[ops_] :=
  With[{u = UnderTheStaircase[ops]},
   If[u === Infinity, Throw["Not holonomic"], Length[u]]];
(* Deduce coefficient list of an OrePolynomial w.r.t. the given support. *)
mycfl[op_, uts_] := Total[Cases[op[[1]], {a_, LeadingExponent[#]} :> a]] & /@ uts;
(* Convert a holonomic system into a Pfaffian
 system (linear system of differential equations). *)
PfaffianSystem[ann_, op_] :=
  With[{uts = UnderTheStaircase[ann]},
   mycfl[OreReduce[op ** #, Together[ann]], uts] & /@ uts];
(* Support for handling expressions in dB. *)
dB[x_] := 10^(x / 10);
```

SNR moment generating function (m.g.f.)

```
(* Write M_{n_1}(s) as defined in Eq.(55). Upon changing index n_1 into m,
M_m corresponds to H_m in Eq.(67). *)
Mm = Sum[Binomial[n1, m1] * (-1)^m1 / (1 - s * Γ1)^(N + n1 - m1), {m1, 0, n1}] /. n1 → m
```

$(s\,\Gamma_1)^m\,(1 - s\,\Gamma_1)^{-m-N}$

```
(* Write the factor that appears in the definition of G_n(z) in Eq.(67),
except H_m. Note that this factor is used several times below. *)
TraditionalForm[
 facGn = Binomial[n, m] * Pochhammer[N, m] / Pochhammer[NR + n - m, m] * c1^m]
```

$$\frac{c_1^m \binom{n}{m} (N)_m}{(-m + n + N_R)_m}$$

```
(* Deduce the annihilator of G_n(z) from Eq.(67),
written for the SNR m.g.f. *)
(* An annihilator is a set of partial differential
 equations (PDEs)/recurrences, in operator notation. *)
(* Note how creative telescoping is employed to deal with this summation. *)
annGn1 = CreativeTelescoping[facGn * Mm, S[m] - 1, {S[n], Der[s], Der[z]}][[1]]
```

$$\Big\{D_z,\ (-n - N_R)\,S_n + \left(-s + s^2\,\Gamma_1 + s^2\,c_1\,\Gamma_1\right) D_s + (n + N_R + N\,s\,\Gamma_1 + N\,s\,c_1\,\Gamma_1),$$
$$\left(-s + 2\,s^2\,\Gamma_1 + s^2\,c_1\,\Gamma_1 - s^3\,\Gamma_1^2 - s^3\,c_1\,\Gamma_1^2\right) D_s^2 + \left(-1 + n + N_R + 3\,s\,\Gamma_1 - n\,s\,\Gamma_1 + 2\,N\,s\,\Gamma_1 + s\,c_1\,\Gamma_1 - \right.$$
$$\left. n\,s\,c_1\,\Gamma_1 + N\,s\,c_1\,\Gamma_1 - s\,N_R\,\Gamma_1 - 2\,s^2\,\Gamma_1^2 - 2\,N\,s^2\,\Gamma_1^2 - 2\,s^2\,c_1\,\Gamma_1^2 - 2\,N\,s^2\,c_1\,\Gamma_1^2\right) D_s +$$
$$\left(N\,\Gamma_1 - n\,N\,\Gamma_1 - n\,N\,c_1\,\Gamma_1 - N\,N_R\,\Gamma_1 - N\,s\,\Gamma_1^2 - N^2\,s\,\Gamma_1^2 - N\,s\,c_1\,\Gamma_1^2 - N^2\,s\,c_1\,\Gamma_1^2\right)\Big\}$$

```
(* Derive the annihilator of the SNR m.g.f.,
by multiplying the remaining terms from Eq.(67), and summing w.r.t. n. Again
 creative telescoping is employed to deal with the summation. *)
annMGF = CreativeTelescoping[DFiniteTimes[annGn1,
    Annihilator[Exp[-z] * z^n / n!, {S[n], Der[s], Der[z]}]], S[n] - 1][[1]]
```

$$\begin{aligned}\{&z D_z^2 + \left(s - s^2 \Gamma_1 - s^2 c_1 \Gamma_1\right) D_s + (z + N_R)\, D_z + (-N s \Gamma_1 - N s c_1 \Gamma_1),\\
&(-z + s z \Gamma_1)\, D_s D_z + (1 - z - N_R - s \Gamma_1 + s z \Gamma_1 + s z c_1 \Gamma_1 + s N_R \Gamma_1)\, D_s +\\
&N z \Gamma_1 D_z + (-N \Gamma_1 + N z \Gamma_1 + N z c_1 \Gamma_1 + N N_R \Gamma_1),\\
&\left(s - 3 s^2 \Gamma_1 - s^2 c_1 \Gamma_1 + 3 s^3 \Gamma_1^2 + 2 s^3 c_1 \Gamma_1^2 - s^4 \Gamma_1^3 - s^4 c_1 \Gamma_1^3\right) D_s^2 +\\
&\left(-2 s \Gamma_1 - 2 N s \Gamma_1 - N s c_1 \Gamma_1 - s z c_1 \Gamma_1 - s c_1 N_R \Gamma_1 + 4 s^2 \Gamma_1^2 + 4 N s^2 \Gamma_1^2 + 2 s^2 c_1 \Gamma_1^2 + 3 N s^2 c_1\right.\\
&\quad\left.\Gamma_1^2 + s^2 z c_1 \Gamma_1^2 + s^2 z c_1^2 \Gamma_1^2 + s^2 c_1 N_R \Gamma_1^2 - 2 s^3 \Gamma_1^3 - 2 N s^3 \Gamma_1^3 - 2 s^3 c_1 \Gamma_1^3 - 2 N s^3 c_1 \Gamma_1^3\right) D_s +\\
&N z c_1 \Gamma_1 D_z + \left(N s \Gamma_1^2 + N^2 s \Gamma_1^2 + N^2 s c_1 \Gamma_1^2 + N s z c_1 \Gamma_1^2 + N s z c_1^2 \Gamma_1^2 +\right.\\
&\quad\left.N s c_1 N_R \Gamma_1^2 - N s^2 \Gamma_1^3 - N^2 s^2 \Gamma_1^3 - N s^2 c_1 \Gamma_1^3 - N^2 s^2 c_1 \Gamma_1^3\right)\}\end{aligned}$$

```
(* Deduce the size of the Pfaffian systems for the m.g.f. *)
HolonomicRank[annMGF]
```

3

```
(* Deduce the vector of derivatives of m(s; z),
which is shown in Section V.C., Step (1). *)
UnderTheStaircase[annMGF]
```

$\{1, D_z, D_s\}$

```
(* From the m.g.f. annihilator,
display the matrices required in Eq.(72), as lists of elements. *)
PSMGFs = PfaffianSystem[annMGF, Der[s]]
PSMGFz = PfaffianSystem[annMGF, Der[z]]
(* From the m.g.f. annihilator, display the matrices required in Eq.(72). *)
Θs == MatrixForm[PSMGFs]
Θz == MatrixForm[PSMGFz]
```

$$\begin{aligned}\Big\{&\{0, 0, 1\}, \Big\{\frac{N \Gamma_1 - N z \Gamma_1 - N z c_1 \Gamma_1 - N N_R \Gamma_1}{z\,(-1 + s \Gamma_1)}, -\frac{N \Gamma_1}{-1 + s \Gamma_1},\\
&(-1 + z + N_R + s \Gamma_1 - s z \Gamma_1 - s z c_1 \Gamma_1 - s N_R \Gamma_1) \,/\, (z\,(-1 + s \Gamma_1))\Big\},\\
&\Big\{\left(N \Gamma_1^2 + N^2 \Gamma_1^2 + N^2 c_1 \Gamma_1^2 + N z c_1 \Gamma_1^2 + N z c_1^2 \Gamma_1^2 + N c_1 N_R \Gamma_1^2 - N s \Gamma_1^3 - N^2 s \Gamma_1^3 - N s c_1 \Gamma_1^3 -\right.\\
&\quad\left.N^2 s c_1 \Gamma_1^3\right) \Big/ \left((-1 + s \Gamma_1)^2\,(-1 + s \Gamma_1 + s c_1 \Gamma_1)\right), \frac{N z c_1 \Gamma_1}{s\,(-1 + s \Gamma_1)^2\,(-1 + s \Gamma_1 + s c_1 \Gamma_1)},\\
&\left(-2 \Gamma_1 - 2 N \Gamma_1 - N c_1 \Gamma_1 - z c_1 \Gamma_1 - c_1 N_R \Gamma_1 + 4 s \Gamma_1^2 + 4 N s \Gamma_1^2 + 2 s c_1 \Gamma_1^2 + 3 N s c_1 \Gamma_1^2 +\right.\\
&\quad\left.s z c_1 \Gamma_1^2 + s z c_1^2 \Gamma_1^2 + s c_1 N_R \Gamma_1^2 - 2 s^2 \Gamma_1^3 - 2 N s^2 \Gamma_1^3 - 2 s^2 c_1 \Gamma_1^3 - 2 N s^2 c_1 \Gamma_1^3\right) \Big/\\
&\left((-1 + s \Gamma_1)^2\,(-1 + s \Gamma_1 + s c_1 \Gamma_1)\right)\Big\}\Big\}\end{aligned}$$

$$\begin{aligned}\Big\{&\{0, 1, 0\}, \Big\{\frac{N s \Gamma_1 + N s c_1 \Gamma_1}{z}, \frac{-z - N_R}{z}, \frac{-s + s^2 \Gamma_1 + s^2 c_1 \Gamma_1}{z}\Big\},\\
&\Big\{\frac{N \Gamma_1 - N z \Gamma_1 - N z c_1 \Gamma_1 - N N_R \Gamma_1}{z\,(-1 + s \Gamma_1)}, -\frac{N \Gamma_1}{-1 + s \Gamma_1},\\
&(-1 + z + N_R + s \Gamma_1 - s z \Gamma_1 - s z c_1 \Gamma_1 - s N_R \Gamma_1) \,/\, (z\,(-1 + s \Gamma_1))\Big\}\Big\}\end{aligned}$$

$$\Theta_s == \begin{pmatrix} 0 & 0 & \\ \frac{N \Gamma_1 - N z \Gamma_1 - N z c_1 \Gamma_1 - N N_R \Gamma_1}{z\,(-1+s \Gamma_1)} & -\frac{N \Gamma_1}{-1+s \Gamma_1} & \\ \frac{N \Gamma_1^2 + N^2 \Gamma_1^2 + N^2 c_1 \Gamma_1^2 + N z c_1 \Gamma_1^2 + N z c_1^2 \Gamma_1^2 + N c_1 N_R \Gamma_1^2 - N s \Gamma_1^3 - N^2 s \Gamma_1^3 - N s c_1 \Gamma_1^3 - N^2 s c_1 \Gamma_1^3}{(-1+s \Gamma_1)^2\,(-1+s \Gamma_1+s c_1 \Gamma_1)} & \frac{N z c_1 \Gamma_1}{s\,(-1+s \Gamma_1)^2\,(-1+s \Gamma_1+s c_1 \Gamma_1)} & -2 \Gamma_1 \cdots \end{pmatrix}$$

$$\Theta_z == \begin{pmatrix} 0 & 1 & 0 \\ \frac{N s \Gamma_1 + N s c_1 \Gamma_1}{z} & \frac{-z-N_R}{z} & \frac{-s+s^2 \Gamma_1+s^2 c_1 \Gamma_1}{z} \\ \frac{N \Gamma_1 - N z \Gamma_1 - N z c_1 \Gamma_1 - N N_R \Gamma_1}{z\,(-1+s \Gamma_1)} & -\frac{N \Gamma_1}{-1+s \Gamma_1} & \frac{-1+z+N_R+s \Gamma_1-s z \Gamma_1-s z c_1 \Gamma_1-s N_R \Gamma_1}{z\,(-1+s \Gamma_1)} \end{pmatrix}$$

SNR probability density function (p.d.f.)

```
(* Write p_{n_1}(t) as defined in Eq.(58). Then,
change index n_1 into m, so that p_{m}(t) corresponds to H_m from
 Eq.(67). The fact that Mathematica outputs p_m(t) as a closed
 form in terms of HypergeometricU for us can be disregarded. *)
pm = Sum[Binomial[n1, m1] * (-1)^m1 * t^(N + n1 - m1 - 1) *
     Exp[-t / Γ1] / (N + n1 - m1 - 1)! / Γ1^(N + n1 - m1), {m1, 0, n1}] /. n1 → m
```

$$\left(e^{-\frac{t}{\Gamma_1}}\, t^{-1+N}\, \text{HypergeometricU}\left[-m, N, \frac{t}{\Gamma_1}\right] \Gamma_1^{-N}\right) \Big/ (-1+m+N)!$$

```
(* Deduce the annihilator of G_n(z) from Eq.(67),
written for the SNR p.d.f. *)
annGn2 = CreativeTelescoping[facGn * pm, S[m] - 1, {S[n], Der[t], Der[z]}][[1]]
```

$$\{D_z,\ (t\,\Gamma_1 + t\,c_1\,\Gamma_1)\,D_t^2 + (-n - N_R)\,S_n + (t + 2\,\Gamma_1 - N\,\Gamma_1 + 2\,c_1\,\Gamma_1 - N\,c_1\,\Gamma_1)\,D_t + (1 + n + N_R),$$
$$(n\,t\,\Gamma_1 + t\,N_R\,\Gamma_1)\,S_n\,D_t + (n\,t + t\,N_R + n\,\Gamma_1 - n\,N\,\Gamma_1 + N_R\,\Gamma_1 - N\,N_R\,\Gamma_1)\,S_n + (t\,c_1\,\Gamma_1 + n\,t\,c_1\,\Gamma_1)\,D_t +$$
$$(c_1\,\Gamma_1 + n\,c_1\,\Gamma_1 - N\,c_1\,\Gamma_1 - n\,N\,c_1\,\Gamma_1),\ (-n\,\Gamma_1 - n^2\,\Gamma_1 - N_R\,\Gamma_1 - 2\,n\,N_R\,\Gamma_1 - N_R^2\,\Gamma_1)\,S_n^2 +$$
$$(n\,t\,c_1 + t\,c_1\,N_R + n\,\Gamma_1 + n^2\,\Gamma_1 - n\,c_1\,\Gamma_1 - n^2\,c_1\,\Gamma_1 - n\,N\,c_1\,\Gamma_1 + N_R\,\Gamma_1 + 2\,n\,N_R\,\Gamma_1 - c_1\,N_R\,\Gamma_1 -$$
$$n\,c_1\,N_R\,\Gamma_1 - N\,c_1\,N_R\,\Gamma_1 + N_R^2\,\Gamma_1)\,S_n + (t\,c_1\,\Gamma_1 + n\,t\,c_1\,\Gamma_1 + t\,c_1^2\,\Gamma_1 + n\,t\,c_1^2\,\Gamma_1)\,D_t +$$
$$(c_1\,\Gamma_1 + 2\,n\,c_1\,\Gamma_1 + n^2\,c_1\,\Gamma_1 + c_1^2\,\Gamma_1 + n\,c_1^2\,\Gamma_1 - N\,c_1^2\,\Gamma_1 - n\,N\,c_1^2\,\Gamma_1 + c_1\,N_R\,\Gamma_1 + n\,c_1\,N_R\,\Gamma_1)\}$$

```
(* The annihilator for the SNR p.d.f., using Eq.(67) as definition. *)
annPDF = CreativeTelescoping[DFiniteTimes[annGn2,
     Annihilator[Exp[-z] * z^n / n!, {S[n], Der[t], Der[z]}]], S[n] - 1][[1]]
```

$$\{t\,z\,\Gamma_1\,D_t\,D_z + (-t\,\Gamma_1 + t\,z\,\Gamma_1 + t\,z\,c_1\,\Gamma_1 + t\,N_R\,\Gamma_1)\,D_t + (t\,z + z\,\Gamma_1 - N\,z\,\Gamma_1)\,D_z +$$
$$(-t + t\,z + t\,N_R - \Gamma_1 + N\,\Gamma_1 + z\,\Gamma_1 - N\,z\,\Gamma_1 + z\,c_1\,\Gamma_1 - N\,z\,c_1\,\Gamma_1 + N_R\,\Gamma_1 - N\,N_R\,\Gamma_1),$$
$$(-t\,\Gamma_1 - t\,c_1\,\Gamma_1)\,D_t^2 + z\,D_z^2 + (-t - 2\,\Gamma_1 + N\,\Gamma_1 - 2\,c_1\,\Gamma_1 + N\,c_1\,\Gamma_1)\,D_t + (z + N_R)\,D_z - 1,$$
$$z^2\,\Gamma_1\,D_z^3 + (2\,z^2\,\Gamma_1 + z^2\,c_1\,\Gamma_1 + 2\,z\,N_R\,\Gamma_1)\,D_z^2 + (-t\,z\,c_1\,\Gamma_1 - t\,z\,c_1^2\,\Gamma_1)\,D_t +$$
$$(-t\,z\,c_1 + z^2\,\Gamma_1 + N\,z\,c_1\,\Gamma_1 + z^2\,c_1\,\Gamma_1 - N_R\,\Gamma_1 + 2\,z\,N_R\,\Gamma_1 + z\,c_1\,N_R\,\Gamma_1 + N_R^2\,\Gamma_1)\,D_z +$$
$$(t\,c_1 - t\,z\,c_1 - t\,c_1\,N_R - N\,c_1\,\Gamma_1 - z\,c_1\,\Gamma_1 + N\,z\,c_1\,\Gamma_1 - z\,c_1^2\,\Gamma_1 + N\,z\,c_1^2\,\Gamma_1 + N\,c_1\,N_R\,\Gamma_1)\}$$

```
(* Alternatively, the annihilator for the SNR p.d.f.,
obtained from the m.g.f. by Laplace transform. *)
(* I.e., multiply the expression with Exp[-s*t] and integrate w.r.t. s. *)
ops = {Der[s], Der[t], Der[z]};
annPDF1 =
 CreativeTelescoping[DFiniteTimes[ToOrePolynomial[Append[annMGF, Der[t]],
     OreAlgebra @@ ops], Annihilator[Exp[-s * t], ops]], Der[s]][[1]]
```

$$\{t\,z\,\Gamma_1\,D_t\,D_z + (-t\,\Gamma_1 + t\,z\,\Gamma_1 + t\,z\,c_1\,\Gamma_1 + t\,N_R\,\Gamma_1)\,D_t + (t\,z + z\,\Gamma_1 - N\,z\,\Gamma_1)\,D_z +$$
$$(-t + t\,z + t\,N_R - \Gamma_1 + N\,\Gamma_1 + z\,\Gamma_1 - N\,z\,\Gamma_1 + z\,c_1\,\Gamma_1 - N\,z\,c_1\,\Gamma_1 + N_R\,\Gamma_1 - N\,N_R\,\Gamma_1),$$
$$(-t\,\Gamma_1 - t\,c_1\,\Gamma_1)\,D_t^2 + z\,D_z^2 + (-t - 2\,\Gamma_1 + N\,\Gamma_1 - 2\,c_1\,\Gamma_1 + N\,c_1\,\Gamma_1)\,D_t + (z + N_R)\,D_z - 1,$$
$$z^2\,\Gamma_1\,D_z^3 + (2\,z^2\,\Gamma_1 + z^2\,c_1\,\Gamma_1 + 2\,z\,N_R\,\Gamma_1)\,D_z^2 + (-t\,z\,c_1\,\Gamma_1 - t\,z\,c_1^2\,\Gamma_1)\,D_t +$$
$$(-t\,z\,c_1 + z^2\,\Gamma_1 + N\,z\,c_1\,\Gamma_1 + z^2\,c_1\,\Gamma_1 - N_R\,\Gamma_1 + 2\,z\,N_R\,\Gamma_1 + z\,c_1\,N_R\,\Gamma_1 + N_R^2\,\Gamma_1)\,D_z +$$
$$(t\,c_1 - t\,z\,c_1 - t\,c_1\,N_R - N\,c_1\,\Gamma_1 - z\,c_1\,\Gamma_1 + N\,z\,c_1\,\Gamma_1 - z\,c_1^2\,\Gamma_1 + N\,z\,c_1^2\,\Gamma_1 + N\,c_1\,N_R\,\Gamma_1)\}$$

```
(* Compare the two results obtained for the p.d.f., they agree. *)
GBEqual[annPDF, annPDF1]
```

True

```
HolonomicRank[annPDF]
```

4

```
(* The vector of derivatives of p(t;
  z) needed for HGM, see Section V.C., Step (2). *)
UnderTheStaircase[annPDF]
```

$$\left\{1,\ D_z,\ D_t,\ D_z^2\right\}$$

```
(* Display the matrices required in the Pfaffian systems from Eq.(73),
as lists of elements. *)
PSPDFt = PfaffianSystem[annPDF, Der[t]]
PSPDFz = PfaffianSystem[annPDF, Der[z]]
(* Display the matrices in matrix form. *)
Ξt == MatrixForm[PSPDFt]
Ξz == MatrixForm[PSPDFz]
```

$$\Big\{\{0,0,1,0\},$$

$$\Big\{\frac{1}{t\,z\,\Gamma_1}\left(t - t\,z - t\,N_R + \Gamma_1 - N\,\Gamma_1 - z\,\Gamma_1 + N\,z\,\Gamma_1 - z\,c_1\,\Gamma_1 + N\,z\,c_1\,\Gamma_1 - N_R\,\Gamma_1 + N\,N_R\,\Gamma_1\right),$$

$$\frac{-t - \Gamma_1 + N\,\Gamma_1}{t\,\Gamma_1},\ \frac{1 - z - z\,c_1 - N_R}{z},\ 0\Big\},$$

$$\Big\{-\frac{1}{t\,(1+c_1)\,\Gamma_1},\ \frac{z + N_R}{t\,(1+c_1)\,\Gamma_1},\ \frac{-t - 2\,\Gamma_1 + N\,\Gamma_1 - 2\,c_1\,\Gamma_1 + N\,c_1\,\Gamma_1}{t\,(1+c_1)\,\Gamma_1},\ \frac{z}{t\,(1+c_1)\,\Gamma_1}\Big\},$$

$$\Big\{\frac{1}{t\,z^2\,\Gamma_1}\Big(-2\,t\,z + t\,z^2 - t\,z\,c_1 + t\,z^2\,c_1 - t\,N_R + 2\,t\,z\,N_R + t\,z\,c_1\,N_R + t\,N_R^2 - 2\,z\,\Gamma_1 + 2\,N\,z\,\Gamma_1 +$$
$$z^2\,\Gamma_1 - N\,z^2\,\Gamma_1 - 2\,z\,c_1\,\Gamma_1 + 2\,N\,z\,c_1\,\Gamma_1 + 2\,z^2\,c_1\,\Gamma_1 - 2\,N\,z^2\,c_1\,\Gamma_1 + z^2\,c_1^2\,\Gamma_1 - N\,z^2\,c_1^2\,\Gamma_1 -$$
$$N_R\,\Gamma_1 + N\,N_R\,\Gamma_1 + 2\,z\,N_R\,\Gamma_1 - 2\,N\,z\,N_R\,\Gamma_1 + 2\,z\,c_1\,N_R\,\Gamma_1 - 2\,N\,z\,c_1\,N_R\,\Gamma_1 + N_R^2\,\Gamma_1 - N\,N_R^2\,\Gamma_1\Big),\ \frac{c_1}{\Gamma_1},$$

$$\frac{1}{z^2}\left(-2\,z + z^2 - 2\,z\,c_1 + 2\,z^2\,c_1 + z^2\,c_1^2 - N_R + 2\,z\,N_R + 2\,z\,c_1\,N_R + N_R^2\right),\ \frac{-t - \Gamma_1 + N\,\Gamma_1}{t\,\Gamma_1}\Big\}\Big\}$$

$$\Big\{\{0,1,0,0\},\ \{0,0,0,1\},$$

$$\Big\{\frac{t - t\,z - t\,N_R + \Gamma_1 - N\,\Gamma_1 - z\,\Gamma_1 + N\,z\,\Gamma_1 - z\,c_1\,\Gamma_1 + N\,z\,c_1\,\Gamma_1 - N_R\,\Gamma_1 + N\,N_R\,\Gamma_1}{t\,z\,\Gamma_1},$$

$$\frac{-t - \Gamma_1 + N\,\Gamma_1}{t\,\Gamma_1},\ \frac{1 - z - z\,c_1 - N_R}{z},\ 0\Big\},$$

$$\Big\{\frac{-t\,c_1 + t\,z\,c_1 + t\,c_1\,N_R + N\,c_1\,\Gamma_1 + z\,c_1\,\Gamma_1 - N\,z\,c_1\,\Gamma_1 + z\,c_1^2\,\Gamma_1 - N\,z\,c_1^2\,\Gamma_1 - N\,c_1\,N_R\,\Gamma_1}{z^2\,\Gamma_1},$$

$$\frac{t\,z\,c_1 - z^2\,\Gamma_1 - N\,z\,c_1\,\Gamma_1 - z^2\,c_1\,\Gamma_1 + N_R\,\Gamma_1 - 2\,z\,N_R\,\Gamma_1 - z\,c_1\,N_R\,\Gamma_1 - N_R^2\,\Gamma_1}{z^2\,\Gamma_1},$$

$$\frac{t\,c_1 + t\,c_1^2}{z},\ \frac{-2\,z - z\,c_1 - 2\,N_R}{z}\Big\}\Big\}$$

$$\Xi_t == \begin{pmatrix} 0 \\ \frac{t-t\,z-t\,N_R+\Gamma_1-N\,\Gamma_1-z\,\Gamma_1+N\,z\,\Gamma_1-z\,c_1\,\Gamma_1+N\,z\,c_1\,\Gamma_1 \ldots}{t\,z\,\Gamma_1} \\ -\frac{1}{t\,(1+c_1)\,\Gamma_1} \\ \frac{-2\,t\,z+t\,z^2-t\,z\,c_1+t\,z^2\,c_1-t\,N_R+2\,t\,z\,N_R+t\,z\,c_1\,N_R+t\,N_R^2-2\,z\,\Gamma_1+2\,N\,z\,\Gamma_1+z^2\,\Gamma_1-N\,z^2\,\Gamma_1-2\,z\,c_1\,\Gamma_1+2\,N\,z\,c_1\,\Gamma_1+2\,z^2\,c_1\,\Gamma_1-2\,N\,z^2\,c_1\,\Gamma_1 \ldots}{t\,z^2\,\Gamma_1} \end{pmatrix}$$

$$\Xi_z == \begin{pmatrix} 0 & 1 \\ 0 & 0 \\ \frac{t-t\,z-t\,N_R+\Gamma_1-N\,\Gamma_1-z\,\Gamma_1+N\,z\,\Gamma_1-z\,c_1\,\Gamma_1+N\,z\,c_1\,\Gamma_1-N_R\,\Gamma_1+N\,N_R\,\Gamma_1}{t\,z\,\Gamma_1} & \frac{-t-\Gamma_1+N\,\Gamma_1}{t\,\Gamma_1} \\ \frac{-t\,c_1+t\,z\,c_1+t\,c_1\,N_R+N\,c_1\,\Gamma_1+z\,c_1\,\Gamma_1-N\,z\,c_1\,\Gamma_1+z\,c_1^2\,\Gamma_1-N\,z\,c_1^2\,\Gamma_1-N\,c_1\,N_R\,\Gamma_1}{z^2\,\Gamma_1} & \frac{t\,z\,c_1-z^2\,\Gamma_1-N\,z\,c_1\,\Gamma_1-z^2\,c_1\,\Gamma_1+N_R\,\Gamma_1-2\,z\,N_R\,\Gamma_1-z\,c \ldots}{z^2\,\Gamma_1} \end{pmatrix}$$

Outage Probability

```
(* Note that our Eq.(63) involves the incomplete gamma function γ(k,x)
 defined as in Footnote 7 on page 7. On the other hand, Mathematica'
 s Gamma[k,x] function is different as revealed by the following: *)
Integrate[t^(k - 1) * Exp[-t], {t, 0, x}, Assumptions → k > 0]
```

```
Gamma[k] - Gamma[k, x]
```

```
(* Write the summand in P_{o,n_1} defined in Eq.(60), by using Eq.(64)
 and by accounting for the Mathematica definition of Gamma[k,x]. *)
TraditionalForm[Pon1Smnd =
  Binomial[n1, m1] * (-1)^m1 * (1 - Gamma[N + n1 - m1, τ / Γ1] / Gamma[N + n1 - m1])]
```

$$(-1)^{m_1}\binom{n_1}{m_1}\left(1-\frac{\Gamma\left(N-m_1+n_1,\frac{\tau}{\Gamma_1}\right)}{\Gamma(N-m_1+n_1)}\right)$$

```
(* Derive the annihilator of P_{o,n_1} w.r.t. n_1. Note
 that the result is a difference operator, which corresponds to
 a difference equation w.r.t. n_1 satisfied by P_{o,n_1}. *)
Annihilator[Sum[Pon1Smnd, {m1, 0, n1}], S[n1]]
```

$$\left\{\left(-\Gamma_1-N\,\Gamma_1-n_1\,\Gamma_1\right)S_{n_1}^2+\left(\tau-\Gamma_1-N\,\Gamma_1-2\,n_1\,\Gamma_1\right)S_{n_1}-n_1\,\Gamma_1\right\}$$

```
(* Note that the above command internally uses
 creative telescoping to derive the annihilator. *)
(* Next, we show an alternative approach that explicitly employs
 creative telescoping; Finally, upon replacing n_1 with m, for Eq.(67),
we obtain the annihilator of H_m (corresponding to P_{o,m}) from
 Eq.(67). Note that this is again a difference operator. *)
annPom = CreativeTelescoping[Pon1Smnd, S[m1] - 1, S[n1]][[1]] /. n1 → m
```

$$\left\{\left(-\Gamma_1-m\,\Gamma_1-N\,\Gamma_1\right)S_m^2+\left(\tau-\Gamma_1-2\,m\,\Gamma_1-N\,\Gamma_1\right)S_m-m\,\Gamma_1\right\}$$

```
(* Define the operators used below. *)
ops = {S[m], S[n], Der[z]};
(* First, derive the annihilator for the summand of G_n in Eq.(67),
written for the outage probability. *)
annGnSmnd = DFiniteTimes[ToOrePolynomial[
    Join[annPom, {S[n] - 1, Der[z]}], OreAlgebra @@ ops], Annihilator[facGn, ops]];
(* Then, do summation w.r.t. m, using a special command
 with a special option. *)
annGn3 = FindCreativeTelescoping[annGnSmnd, S[m] - 1,
   Denominator → (n + N_R) * Pochhammer[n - m, 5]][[1]]
```

$$
\begin{aligned}
\{&D_z,\\
&(6\,\Gamma_1 + 11\,n\,\Gamma_1 + 6\,n^2\,\Gamma_1 + n^3\,\Gamma_1 + 11\,N_R\,\Gamma_1 + 12\,n\,N_R\,\Gamma_1 + 3\,n^2\,N_R\,\Gamma_1 + 6\,N_R^2\,\Gamma_1 + 3\,n\,N_R^2\,\Gamma_1 + N_R^3\,\Gamma_1)\,S_n^4 +\\
&(-2\,\tau\,c_1 - 3\,n\,\tau\,c_1 - n^2\,\tau\,c_1 - 3\,\tau\,c_1\,N_R - 2\,n\,\tau\,c_1\,N_R - \tau\,c_1\,N_R^2 - 8\,\Gamma_1 - 16\,n\,\Gamma_1 - 10\,n^2\,\Gamma_1 - 2\,n^3\,\Gamma_1 +\\
&\quad 10\,c_1\,\Gamma_1 + 19\,n\,c_1\,\Gamma_1 + 11\,n^2\,c_1\,\Gamma_1 + 2\,n^3\,c_1\,\Gamma_1 + 2\,N\,c_1\,\Gamma_1 + 3\,n\,N\,c_1\,\Gamma_1 + n^2\,N\,c_1\,\Gamma_1 -\\
&\quad 16\,N_R\,\Gamma_1 - 20\,n\,N_R\,\Gamma_1 - 6\,n^2\,N_R\,\Gamma_1 + 15\,c_1\,N_R\,\Gamma_1 + 16\,n\,c_1\,N_R\,\Gamma_1 + 4\,n^2\,c_1\,N_R\,\Gamma_1 + 3\,N\,c_1\,N_R\,\Gamma_1 +\\
&\quad 2\,n\,N\,c_1\,N_R\,\Gamma_1 - 10\,N_R^2\,\Gamma_1 - 6\,n\,N_R^2\,\Gamma_1 + 5\,c_1\,N_R^2\,\Gamma_1 + 2\,n\,c_1\,N_R^2\,\Gamma_1 + N\,c_1\,N_R^2\,\Gamma_1 - 2\,N_R^3\,\Gamma_1)\,S_n^3 +\\
&(4\,\tau\,c_1 + 6\,n\,\tau\,c_1 + 2\,n^2\,\tau\,c_1 + 5\,\tau\,c_1\,N_R + 3\,n\,\tau\,c_1\,N_R + \tau\,c_1\,N_R^2 + 2\,\Gamma_1 + 5\,n\,\Gamma_1 + 4\,n^2\,\Gamma_1 +\\
&\quad n^3\,\Gamma_1 - 12\,c_1\,\Gamma_1 - 26\,n\,c_1\,\Gamma_1 - 18\,n^2\,c_1\,\Gamma_1 - 4\,n^3\,c_1\,\Gamma_1 - 4\,N\,c_1\,\Gamma_1 - 6\,n\,N\,c_1\,\Gamma_1 -\\
&\quad 2\,n^2\,N\,c_1\,\Gamma_1 + 4\,c_1^2\,\Gamma_1 + 8\,n\,c_1^2\,\Gamma_1 + 5\,n^2\,c_1^2\,\Gamma_1 + n^3\,c_1^2\,\Gamma_1 + 2\,N\,c_1^2\,\Gamma_1 + 3\,n\,N\,c_1^2\,\Gamma_1 +\\
&\quad n^2\,N\,c_1^2\,\Gamma_1 + 5\,N_R\,\Gamma_1 + 8\,n\,N_R\,\Gamma_1 + 3\,n^2\,N_R\,\Gamma_1 - 21\,c_1\,N_R\,\Gamma_1 - 27\,n\,c_1\,N_R\,\Gamma_1 - 8\,n^2\,c_1\,N_R\,\Gamma_1 -\\
&\quad 5\,N\,c_1\,N_R\,\Gamma_1 - 3\,n\,N\,c_1\,N_R\,\Gamma_1 + 4\,c_1^2\,N_R\,\Gamma_1 + 4\,n\,c_1^2\,N_R\,\Gamma_1 + n^2\,c_1^2\,N_R\,\Gamma_1 + 2\,N\,c_1^2\,N_R\,\Gamma_1 +\\
&\quad n\,N\,c_1^2\,N_R\,\Gamma_1 + 4\,N_R^2\,\Gamma_1 + 3\,n\,N_R^2\,\Gamma_1 - 9\,c_1\,N_R^2\,\Gamma_1 - 4\,n\,c_1\,N_R^2\,\Gamma_1 - N\,c_1\,N_R^2\,\Gamma_1 + N_R^3\,\Gamma_1)\,S_n^2 +\\
&(-2\,\tau\,c_1 - 3\,n\,\tau\,c_1 - n^2\,\tau\,c_1 - 2\,\tau\,c_1\,N_R - n\,\tau\,c_1\,N_R + 2\,c_1\,\Gamma_1 + 7\,n\,c_1\,\Gamma_1 + 7\,n^2\,c_1\,\Gamma_1 +\\
&\quad 2\,n^3\,c_1\,\Gamma_1 + 2\,N\,c_1\,\Gamma_1 + 3\,n\,N\,c_1\,\Gamma_1 + n^2\,N\,c_1\,\Gamma_1 - 4\,c_1^2\,\Gamma_1 - 10\,n\,c_1^2\,\Gamma_1 - 8\,n^2\,c_1^2\,\Gamma_1 -\\
&\quad 2\,n^3\,c_1^2\,\Gamma_1 - 4\,N\,c_1^2\,\Gamma_1 - 6\,n\,N\,c_1^2\,\Gamma_1 - 2\,n^2\,N\,c_1^2\,\Gamma_1 + 6\,c_1\,N_R\,\Gamma_1 + 11\,n\,c_1\,N_R\,\Gamma_1 +\\
&\quad 4\,n^2\,c_1\,N_R\,\Gamma_1 + 2\,N\,c_1\,N_R\,\Gamma_1 + n\,N\,c_1\,N_R\,\Gamma_1 - 6\,c_1^2\,N_R\,\Gamma_1 - 7\,n\,c_1^2\,N_R\,\Gamma_1 -\\
&\quad 2\,n^2\,c_1^2\,N_R\,\Gamma_1 - 2\,N\,c_1^2\,N_R\,\Gamma_1 - n\,N\,c_1^2\,N_R\,\Gamma_1 + 4\,c_1\,N_R^2\,\Gamma_1 + 2\,n\,c_1\,N_R^2\,\Gamma_1)\,S_n +\\
&(2\,n\,c_1^2\,\Gamma_1 + 3\,n^2\,c_1^2\,\Gamma_1 + n^3\,c_1^2\,\Gamma_1 + 2\,N\,c_1^2\,\Gamma_1 + 3\,n\,N\,c_1^2\,\Gamma_1 + n^2\,N\,c_1^2\,\Gamma_1 +\\
&\quad 2\,c_1^2\,N_R\,\Gamma_1 + 3\,n\,c_1^2\,N_R\,\Gamma_1 + n^2\,c_1^2\,N_R\,\Gamma_1)\}
\end{aligned}
$$

```
(* Derive the annihilator for the outage probability,
using Eq.(67) as definition. *)
annOP = CreativeTelescoping[DFiniteTimes[annGn3,
    Annihilator[Exp[-z] * z^n / n!, {S[n], Der[z]}]], S[n] - 1][[1]]
```

$$
\begin{aligned}
\{&z^3\,\Gamma_1\,D_z^5 + (3\,z^2\,\Gamma_1 + 3\,z^3\,\Gamma_1 + 2\,z^3\,c_1\,\Gamma_1 + 3\,z^2\,N_R\,\Gamma_1)\,D_z^4 +\\
&(-z^2\,\tau\,c_1 + 8\,z^2\,\Gamma_1 + 3\,z^3\,\Gamma_1 + 5\,z^2\,c_1\,\Gamma_1 + N\,z^2\,c_1\,\Gamma_1 + 4\,z^3\,c_1\,\Gamma_1 +\\
&\quad z^3\,c_1^2\,\Gamma_1 + 3\,z\,N_R\,\Gamma_1 + 6\,z^2\,N_R\,\Gamma_1 + 4\,z^2\,c_1\,N_R\,\Gamma_1 + 3\,z\,N_R^2\,\Gamma_1)\,D_z^3 +\\
&(-z^2\,\tau\,c_1 - 2\,z\,\tau\,c_1\,N_R + 7\,z^2\,\Gamma_1 + z^3\,\Gamma_1 + 9\,z^2\,c_1\,\Gamma_1 + N\,z^2\,c_1\,\Gamma_1 + 2\,z^3\,c_1\,\Gamma_1 +\\
&\quad 2\,z^2\,c_1^2\,\Gamma_1 + N\,z^2\,c_1^2\,\Gamma_1 + z^3\,c_1^2\,\Gamma_1 - N_R\,\Gamma_1 + 7\,z\,N_R\,\Gamma_1 + 3\,z^2\,N_R\,\Gamma_1 + 4\,z\,c_1\,N_R\,\Gamma_1 +\\
&\quad 2\,N\,z\,c_1\,N_R\,\Gamma_1 + 4\,z^2\,c_1\,N_R\,\Gamma_1 + z^2\,c_1^2\,N_R\,\Gamma_1 + 3\,z\,N_R^2\,\Gamma_1 + 2\,z\,c_1\,N_R^2\,\Gamma_1 + N_R^3\,\Gamma_1)\,D_z^2 +\\
&(\tau\,c_1\,N_R - z\,\tau\,c_1\,N_R - \tau\,c_1\,N_R^2 + 2\,z^2\,\Gamma_1 + 4\,z^2\,c_1\,\Gamma_1 + 2\,z^2\,c_1^2\,\Gamma_1 - 2\,N_R\,\Gamma_1 +\\
&\quad 4\,z\,N_R\,\Gamma_1 - c_1\,N_R\,\Gamma_1 - N\,c_1\,N_R\,\Gamma_1 + 5\,z\,c_1\,N_R\,\Gamma_1 + N\,z\,c_1\,N_R\,\Gamma_1 +\\
&\quad z\,c_1^2\,N_R\,\Gamma_1 + N\,z\,c_1^2\,N_R\,\Gamma_1 + 2\,N_R^2\,\Gamma_1 + c_1\,N_R^2\,\Gamma_1 + N\,c_1\,N_R^2\,\Gamma_1)\,D_z\}
\end{aligned}
$$

```
(* Alternatively,
compute the annihilator of the outage probability directly, *)
(* by integrating the SNR p.d.f. from 0 to τ, as in Eq.(59). *)
(* Because of the non-
 natural upper bound we have to consider inhomogeneous parts. *)
ct = CreativeTelescoping[annPDF, Der[t]];
annOP1 = OreGroebnerBasis[
  Flatten[MapThread[Function[{p, q}, (# ** p) & /@ DFiniteSubstitute[
       DFiniteOreAction[annPDF, q], {t → τ}, Algebra → OreAlgebra[Der[z]]]], ct]]]
```

$$\{z^3\,\Gamma_1\,D_z^5 + \left(3\,z^2\,\Gamma_1 + 3\,z^3\,\Gamma_1 + 2\,z^3\,c_1\,\Gamma_1 + 3\,z^2\,N_R\,\Gamma_1\right) D_z^4 + \left(-z^2\,\tau\,c_1 + 8\,z^2\,\Gamma_1 + 3\,z^3\,\Gamma_1 + 5\,z^2\,c_1\,\Gamma_1 + N\,z^2\,c_1\,\Gamma_1 + 4\,z^3\,c_1\,\Gamma_1 + z^3\,c_1^2\,\Gamma_1 + 3\,z\,N_R\,\Gamma_1 + 6\,z^2\,N_R\,\Gamma_1 + 4\,z^2\,c_1\,N_R\,\Gamma_1 + 3\,z\,N_R^2\,\Gamma_1\right) D_z^3 + \left(-z^2\,\tau\,c_1 - 2\,z\,\tau\,c_1\,N_R + 7\,z^2\,\Gamma_1 + z^3\,\Gamma_1 + 9\,z^2\,c_1\,\Gamma_1 + N\,z^2\,c_1\,\Gamma_1 + 2\,z^3\,c_1\,\Gamma_1 + 2\,z^2\,c_1^2\,\Gamma_1 + N\,z^2\,c_1^2\,\Gamma_1 + z^3\,c_1^2\,\Gamma_1 - N_R\,\Gamma_1 + 7\,z\,N_R\,\Gamma_1 + 3\,z^2\,N_R\,\Gamma_1 + 4\,z\,c_1\,N_R\,\Gamma_1 + 2\,N\,z\,c_1\,N_R\,\Gamma_1 + 4\,z^2\,c_1\,N_R\,\Gamma_1 + z^2\,c_1^2\,N_R\,\Gamma_1 + 3\,z\,N_R^2\,\Gamma_1 + 2\,z\,c_1\,N_R^2\,\Gamma_1 + N_R^3\,\Gamma_1\right) D_z^2 + \left(\tau\,c_1\,N_R - z\,\tau\,c_1\,N_R - \tau\,c_1\,N_R^2 + 2\,z^2\,\Gamma_1 + 4\,z^2\,c_1\,\Gamma_1 + 2\,z^2\,c_1^2\,\Gamma_1 - 2\,N_R\,\Gamma_1 + 4\,z\,N_R\,\Gamma_1 - c_1\,N_R\,\Gamma_1 - N\,c_1\,N_R\,\Gamma_1 + 5\,z\,c_1\,N_R\,\Gamma_1 + N\,z\,c_1\,N_R\,\Gamma_1 + z\,c_1^2\,N_R\,\Gamma_1 + N\,z\,c_1^2\,N_R\,\Gamma_1 + 2\,N_R^2\,\Gamma_1 + c_1\,N_R^2\,\Gamma_1 + N\,c_1\,N_R^2\,\Gamma_1\right) D_z\}$$

```
(* Test that the two different methods to
 compute this annihilator yield the same result. *)
GBEqual[annOP, annOP1]
```

True

```
HolonomicRank[annOP]
```

5

```
(* Reveal that annihilator of P_o(z) consists of a
 single ordinary differential equation of order 5 w.r.t. z. *)
Support[annOP]
```

$\{\{D_z^5, D_z^4, D_z^3, D_z^2, D_z\}\}$

```
(* The vector of derivatives needed in HGM. *)
UnderTheStaircase[annOP]
```

$\{1, D_z, D_z^2, D_z^3, D_z^4\}$

```
(* Display the companion matrix Φ of the corresponding
 Pfaffian system from Eq.(74) as list of elements. *)
PSOP = PfaffianSystem[annOP1, Der[z]]
(* Display the companion matrix Φ of the
 corresponding Pfaffian system from Eq.(74). Unfortunately,
the matrix is fully viewable only under Mathematica (.nb),
or under the Wolfram CDF player (.cdf). *)
MatrixForm[PSOP]
```

$$\{\{0, 1, 0, 0, 0\}, \{0, 0, 1, 0, 0\}, \{0, 0, 0, 1, 0\}, \{0, 0, 0, 0, 1\}, \Big\{0, \frac{1}{z^3\,\Gamma_1}\,\big(-\tau\,c_1\,N_R + z\,\tau\,c_1\,N_R + \tau\,c_1\,N_R^2 - 2\,z^2\,\Gamma_1 - 4\,z^2\,c_1\,\Gamma_1 - 2\,z^2\,c_1^2\,\Gamma_1 + 2\,N_R\,\Gamma_1 - 4\,z\,N_R\,\Gamma_1 + c_1\,N_R\,\Gamma_1 + N\,c_1\,N_R\,\Gamma_1 - 5\,z\,c_1\,N_R\,\Gamma_1 - N\,z\,c_1\,N_R\,\Gamma_1 - z\,c_1^2\,N_R\,\Gamma_1 - N\,z\,c_1^2\,N_R\,\Gamma_1 - 2\,N_R^2\,\Gamma_1 - c_1\,N_R^2\,\Gamma_1 - N\,c_1\,N_R^2\,\Gamma_1\big),$$

$$\frac{1}{z^3\,\Gamma_1}\,\big(z^2\,\tau\,c_1 + 2\,z\,\tau\,c_1\,N_R - 7\,z^2\,\Gamma_1 - z^3\,\Gamma_1 - 9\,z^2\,c_1\,\Gamma_1 - N\,z^2\,c_1\,\Gamma_1 - 2\,z^3\,c_1\,\Gamma_1 - 2\,z^2\,c_1^2\,\Gamma_1 - N\,z^2\,c_1^2\,\Gamma_1 - z^3\,c_1^2\,\Gamma_1 + N_R\,\Gamma_1 - 7\,z\,N_R\,\Gamma_1 - 3\,z^2\,N_R\,\Gamma_1 - 4\,z\,c_1\,N_R\,\Gamma_1 - 2\,N\,z\,c_1\,N_R\,\Gamma_1 - 4\,z^2\,c_1\,N_R\,\Gamma_1 - z^2\,c_1^2\,N_R\,\Gamma_1 - 3\,z\,N_R^2\,\Gamma_1 - 2\,z\,c_1\,N_R^2\,\Gamma_1 - N_R^3\,\Gamma_1\big),$$

$$\frac{1}{z^2\,\Gamma_1}\,\big(z\,\tau\,c_1 - 8\,z\,\Gamma_1 - 3\,z^2\,\Gamma_1 - 5\,z\,c_1\,\Gamma_1 - N\,z\,c_1\,\Gamma_1 - 4\,z^2\,c_1\,\Gamma_1 - z^2\,c_1^2\,\Gamma_1 - 3\,N_R\,\Gamma_1 - 6\,z\,N_R\,\Gamma_1 - 4\,z\,c_1\,N_R\,\Gamma_1 - 3\,N_R^2\,\Gamma_1\big),\ \frac{-3 - 3\,z - 2\,z\,c_1 - 3\,N_R}{z}\Big\}\}$$

$$\begin{pmatrix} 0 & 1 \\ 0 & 0 \\ 0 & 0 \\ 0 & 0 \\ 0 & \frac{-\tau\,c_1\,N_R+z\,\tau\,c_1\,N_R+\tau\,c_1\,N_R^2-2\,z^2\,\Gamma_1-4\,z^2\,c_1\,\Gamma_1-2\,z^2\,c_1^2\,\Gamma_1+2\,N_R\,\Gamma_1-4\,z\,N_R\,\Gamma_1+c_1\,N_R\,\Gamma_1+N\,c_1\,N_R\,\Gamma_1-5\,z\,c_1\,N_R\,\Gamma_1-N\,z\,c_1\,N_R\,\Gamma_1-z\,c_1^2\,N_R\,\Gamma_1-N\,z}{z^3\,\Gamma_1} \end{pmatrix}$$

```
(* The following allows us to easily copy-paste the results into Matlab. *)
(*InputForm[PSOP]*)
```

Ergodic Capacity

```
(* Derive the annihilator for the ergodic capacity C(z),
using the definition in Eq.(61). *)
annC = FindCreativeTelescoping[DFiniteTimes[
    Annihilator[Log[2, t + 1], {Der[t], Der[z]}], annPDF], Der[t]][[1]]
```

$$
\begin{aligned}
\{ & z^4 \Gamma_1 D_z^7 + \left(8 z^3 \Gamma_1 + 4 z^4 \Gamma_1 + 2 z^4 c_1 \Gamma_1 + 4 z^3 N_R \Gamma_1\right) D_z^6 + \\
& \left(z^3 c_1 + 12 z^2 \Gamma_1 + 30 z^3 \Gamma_1 + 6 z^4 \Gamma_1 + 15 z^3 c_1 \Gamma_1 + N z^3 c_1 \Gamma_1 + 6 z^4 c_1 \Gamma_1 + \right. \\
& \quad \left. z^4 c_1^2 \Gamma_1 + 18 z^2 N_R \Gamma_1 + 12 z^3 N_R \Gamma_1 + 6 z^3 c_1 N_R \Gamma_1 + 6 z^2 N_R^2 \Gamma_1\right) D_z^5 + \\
& \left(3 z^2 c_1 + 2 z^3 c_1 + 3 z^2 c_1 N_R + 42 z^2 \Gamma_1 + 42 z^3 \Gamma_1 + 4 z^4 \Gamma_1 + 21 z^2 c_1 \Gamma_1 + \right. \\
& \quad 3 N z^2 c_1 \Gamma_1 + 42 z^3 c_1 \Gamma_1 + 2 N z^3 c_1 \Gamma_1 + 6 z^4 c_1 \Gamma_1 + 7 z^3 c_1^2 \Gamma_1 + N z^3 c_1^2 \Gamma_1 + \\
& \quad 2 z^4 c_1^2 \Gamma_1 + 8 z N_R \Gamma_1 + 54 z^2 N_R \Gamma_1 + 12 z^3 N_R \Gamma_1 + 27 z^2 c_1 N_R \Gamma_1 + 3 N z^2 c_1 N_R \Gamma_1 + \\
& \quad \left. 12 z^3 c_1 N_R \Gamma_1 + 2 z^3 c_1^2 N_R \Gamma_1 + 12 z N_R^2 \Gamma_1 + 12 z^2 N_R^2 \Gamma_1 + 6 z^2 c_1 N_R^2 \Gamma_1 + 4 z N_R^3 \Gamma_1\right) D_z^4 + \\
& \left(6 z^2 c_1 + z^3 c_1 + 3 z c_1 N_R + 4 z^2 c_1 N_R + 3 z c_1 N_R^2 + 54 z^2 \Gamma_1 + 26 z^3 \Gamma_1 + z^4 \Gamma_1 + \right. \\
& \quad 54 z^2 c_1 \Gamma_1 + 6 N z^2 c_1 \Gamma_1 + 39 z^3 c_1 \Gamma_1 + N z^3 c_1 \Gamma_1 + 2 z^4 c_1 \Gamma_1 + 9 z^2 c_1^2 \Gamma_1 + 3 N z^2 c_1^2 \Gamma_1 + \\
& \quad 13 z^3 c_1^2 \Gamma_1 + N z^3 c_1^2 \Gamma_1 + z^4 c_1^2 \Gamma_1 - 2 N_R \Gamma_1 + 26 z N_R \Gamma_1 + 54 z^2 N_R \Gamma_1 + 4 z^3 N_R \Gamma_1 + \\
& \quad 13 z c_1 N_R \Gamma_1 + 3 N z c_1 N_R \Gamma_1 + 54 z^2 c_1 N_R \Gamma_1 + 4 N z^2 c_1 N_R \Gamma_1 + 6 z^3 c_1 N_R \Gamma_1 + \\
& \quad 9 z^2 c_1^2 N_R \Gamma_1 + 2 N z^2 c_1^2 N_R \Gamma_1 + 2 z^3 c_1^2 N_R \Gamma_1 - N_R^2 \Gamma_1 + 30 z N_R^2 \Gamma_1 + 6 z^2 N_R^2 \Gamma_1 + 15 z c_1 N_R^2 \Gamma_1 + \\
& \quad \left. 3 N z c_1 N_R^2 \Gamma_1 + 6 z^2 c_1 N_R^2 \Gamma_1 + z^2 c_1^2 N_R^2 \Gamma_1 + 2 N_R^3 \Gamma_1 + 4 z N_R^3 \Gamma_1 + 2 z c_1 N_R^3 \Gamma_1 + N_R^4 \Gamma_1\right) D_z^3 + \\
& \left(3 z^2 c_1 - c_1 N_R + 6 z c_1 N_R + z^2 c_1 N_R + 2 z c_1 N_R^2 + c_1 N_R^3 + 30 z^2 \Gamma_1 + 6 z^3 \Gamma_1 + 45 z^2 c_1 \Gamma_1 + \right. \\
& \quad 3 N z^2 c_1 \Gamma_1 + 12 z^3 c_1 \Gamma_1 + 15 z^2 c_1^2 \Gamma_1 + 3 N z^2 c_1^2 \Gamma_1 + 6 z^3 c_1^2 \Gamma_1 - 6 N_R \Gamma_1 + 30 z N_R \Gamma_1 + \\
& \quad 18 z^2 N_R \Gamma_1 - 3 c_1 N_R \Gamma_1 - N c_1 N_R \Gamma_1 + 30 z c_1 N_R \Gamma_1 + 6 N z c_1 N_R \Gamma_1 + 27 z^2 c_1 N_R \Gamma_1 + \\
& \quad N z^2 c_1 N_R \Gamma_1 + 5 z c_1^2 N_R \Gamma_1 + 3 N z c_1^2 N_R \Gamma_1 + 9 z^2 c_1^2 N_R \Gamma_1 + N z^2 c_1^2 N_R \Gamma_1 + 18 z N_R^2 \Gamma_1 + \\
& \quad \left. 18 z c_1 N_R^2 \Gamma_1 + 2 N z c_1 N_R^2 \Gamma_1 + 3 z c_1^2 N_R^2 \Gamma_1 + N z c_1^2 N_R^2 \Gamma_1 + 6 N_R^3 \Gamma_1 + 3 c_1 N_R^3 \Gamma_1 + N c_1 N_R^3 \Gamma_1\right) D_z^2 + \\
& \left(-2 c_1 N_R + 2 z c_1 N_R + 2 c_1 N_R^2 + 6 z^2 \Gamma_1 + 12 z^2 c_1 \Gamma_1 + 6 z^2 c_1^2 \Gamma_1 - 6 N_R \Gamma_1 + 12 z N_R \Gamma_1 - \right. \\
& \quad 6 c_1 N_R \Gamma_1 - 2 N c_1 N_R \Gamma_1 + 18 z c_1 N_R \Gamma_1 + 2 N z c_1 N_R \Gamma_1 - c_1^2 N_R \Gamma_1 - N c_1^2 N_R \Gamma_1 + \\
& \quad \left. 6 z c_1^2 N_R \Gamma_1 + 2 N z c_1^2 N_R \Gamma_1 + 6 N_R^2 \Gamma_1 + 6 c_1 N_R^2 \Gamma_1 + 2 N c_1 N_R^2 \Gamma_1 + c_1^2 N_R^2 \Gamma_1 + N c_1^2 N_R^2 \Gamma_1\right) D_z \}
\end{aligned}
$$

```
(* Reveal that C(z) satisfies an
 ordinary differential equation of order 7. *)
Support[annC]
```

$$\left\{\left\{D_z^7, D_z^6, D_z^5, D_z^4, D_z^3, D_z^2, D_z\right\}\right\}$$

```
(* The vector of derivatives needed for HGM. *)
UnderTheStaircase[annC]
```

$$\left\{1, D_z, D_z^2, D_z^3, D_z^4, D_z^5, D_z^6\right\}$$

```
(* Display the companion matrix Ψ of the corresponding Pfaffian system,
Eq.(74), as a list of elements. *)
PSC = PfaffianSystem[annC, Der[z]]
(* Display the companion matrix Ψ of the corresponding Pfaffian system,
Eq.(74). Unfortunately,
the matrix is fully viewable only under Mathematica (.nb),
or under the Wolfram CDF player (.cdf). *)
MatrixForm[PSC]
```

$$\{\{0, 1, 0, 0, 0, 0, 0\}, \{0, 0, 1, 0, 0, 0, 0\}, \{0, 0, 0, 1, 0, 0, 0\}, \{0, 0, 0, 0, 1, 0, 0\}, \{0, 0, 0, 0, 0, 1, 0\}, \{0, 0, 0, 0, 0, 0, 1\},$$

$$\{0, \frac{1}{z^4 \Gamma_1} \big(2 c_1 N_R - 2 z c_1 N_R - 2 c_1 N_R^2 - 6 z^2 \Gamma_1 - 12 z^2 c_1 \Gamma_1 - 6 z^2 c_1^2 \Gamma_1 + 6 N_R \Gamma_1 - 12 z N_R \Gamma_1 + 6 c_1 N_R \Gamma_1 + 2 N c_1 N_R \Gamma_1 - 18 z c_1 N_R \Gamma_1 - 2 N z c_1 N_R \Gamma_1 + c_1^2 N_R \Gamma_1 + N c_1^2 N_R \Gamma_1 - 6 z c_1^2 N_R \Gamma_1 - 2 N z c_1^2 N_R \Gamma_1 - 6 N_R^2 \Gamma_1 - 6 c_1 N_R^2 \Gamma_1 - 2 N c_1 N_R^2 \Gamma_1 - c_1^2 N_R^2 \Gamma_1 - N c_1^2 N_R^2 \Gamma_1\big),$$

$$\frac{1}{z^4 \Gamma_1} \big(-3 z^2 c_1 + c_1 N_R - 6 z c_1 N_R - z^2 c_1 N_R - 2 z c_1 N_R^2 - c_1 N_R^3 - 30 z^2 \Gamma_1 - 6 z^3 \Gamma_1 - 45 z^2 c_1 \Gamma_1 - 3 N z^2 c_1 \Gamma_1 - 12 z^3 c_1 \Gamma_1 - 15 z^2 c_1^2 \Gamma_1 - 3 N z^2 c_1^2 \Gamma_1 - 6 z^3 c_1^2 \Gamma_1 + 6 N_R \Gamma_1 - 30 z N_R \Gamma_1 - 18 z^2 N_R \Gamma_1 + 3 c_1 N_R \Gamma_1 + N c_1 N_R \Gamma_1 - 30 z c_1 N_R \Gamma_1 - 6 N z c_1 N_R \Gamma_1 - 27 z^2 c_1 N_R \Gamma_1 - N z^2 c_1 N_R \Gamma_1 - 5 z c_1^2 N_R \Gamma_1 - 3 N z c_1^2 N_R \Gamma_1 - 9 z^2 c_1^2 N_R \Gamma_1 - N z^2 c_1^2 N_R \Gamma_1 - 18 z N_R^2 \Gamma_1 - 18 z c_1 N_R^2 \Gamma_1 - 2 N z c_1 N_R^2 \Gamma_1 - 3 z c_1^2 N_R^2 \Gamma_1 - N z c_1^2 N_R^2 \Gamma_1 - 6 N_R^3 \Gamma_1 - 3 c_1 N_R^3 \Gamma_1 - N c_1 N_R^3 \Gamma_1\big),$$

$$\frac{1}{z^4 \Gamma_1} \big(-6 z^2 c_1 - z^3 c_1 - 3 z c_1 N_R - 4 z^2 c_1 N_R - 3 z c_1 N_R^2 - 54 z^2 \Gamma_1 - 26 z^3 \Gamma_1 - z^4 \Gamma_1 - 54 z^2 c_1 \Gamma_1 - 6 N z^2 c_1 \Gamma_1 - 39 z^3 c_1 \Gamma_1 - N z^3 c_1 \Gamma_1 - 2 z^4 c_1 \Gamma_1 - 9 z^2 c_1^2 \Gamma_1 - 3 N z^2 c_1^2 \Gamma_1 - 13 z^3 c_1^2 \Gamma_1 - N z^3 c_1^2 \Gamma_1 - z^4 c_1^2 \Gamma_1 + 2 N_R \Gamma_1 - 26 z N_R \Gamma_1 - 54 z^2 N_R \Gamma_1 - 4 z^3 N_R \Gamma_1 - 13 z c_1 N_R \Gamma_1 - 3 N z c_1 N_R \Gamma_1 - 54 z^2 c_1 N_R \Gamma_1 - 4 N z^2 c_1 N_R \Gamma_1 - 6 z^3 c_1 N_R \Gamma_1 - 9 z^2 c_1^2 N_R \Gamma_1 - 2 N z^2 c_1^2 N_R \Gamma_1 - 2 z^3 c_1^2 N_R \Gamma_1 + N_R^2 \Gamma_1 - 30 z N_R^2 \Gamma_1 - 6 z^2 N_R^2 \Gamma_1 - 15 z c_1 N_R^2 \Gamma_1 - 3 N z c_1 N_R^2 \Gamma_1 - 6 z^2 c_1 N_R^2 \Gamma_1 - z^2 c_1^2 N_R^2 \Gamma_1 - 2 N_R^3 \Gamma_1 - 4 z N_R^3 \Gamma_1 - 2 z c_1 N_R^3 \Gamma_1 - N_R^4 \Gamma_1\big),$$

$$\frac{1}{z^3 \Gamma_1} \big(-3 z c_1 - 2 z^2 c_1 - 3 z c_1 N_R - 42 z \Gamma_1 - 42 z^2 \Gamma_1 - 4 z^3 \Gamma_1 - 21 z c_1 \Gamma_1 - 3 N z c_1 \Gamma_1 - 42 z^2 c_1 \Gamma_1 - 2 N z^2 c_1 \Gamma_1 - 6 z^3 c_1 \Gamma_1 - 7 z^2 c_1^2 \Gamma_1 - N z^2 c_1^2 \Gamma_1 - 2 z^3 c_1^2 \Gamma_1 - 8 N_R \Gamma_1 - 54 z N_R \Gamma_1 - 12 z^2 N_R \Gamma_1 - 27 z c_1 N_R \Gamma_1 - 3 N z c_1 N_R \Gamma_1 - 12 z^2 c_1 N_R \Gamma_1 - 2 z^2 c_1^2 N_R \Gamma_1 - 12 N_R^2 \Gamma_1 - 12 z N_R^2 \Gamma_1 - 6 z c_1 N_R^2 \Gamma_1 - 4 N_R^3 \Gamma_1\big),$$

$$\frac{1}{z^2 \Gamma_1} \big(-z c_1 - 12 \Gamma_1 - 30 z \Gamma_1 - 6 z^2 \Gamma_1 - 15 z c_1 \Gamma_1 - N z c_1 \Gamma_1 - 6 z^2 c_1 \Gamma_1 - z^2 c_1^2 \Gamma_1 - 18 N_R \Gamma_1 - 12 z N_R \Gamma_1 - 6 z c_1 N_R \Gamma_1 - 6 N_R^2 \Gamma_1\big), -\frac{2 (4 + 2 z + z c_1 + 2 N_R)}{z}\}\}$$

$$\left(\begin{array}{ll} 0 & 1 \\ 0 & 0 \\ 0 & 0 \\ 0 & 0 \\ 0 & 0 \\ 0 & 0 \\ 0 & \frac{2 c_1 N_R - 2 z c_1 N_R - 2 c_1 N_R^2 - 6 z^2 \Gamma_1 - 12 z^2 c_1 \Gamma_1 - 6 z^2 c_1^2 \Gamma_1 + 6 N_R \Gamma_1 - 12 z N_R \Gamma_1 + 6 c_1 N_R \Gamma_1 + 2 N c_1 N_R \Gamma_1 - 18 z c_1 N_R \Gamma_1 - 2 N z c_1 N_R \Gamma_1 + c_1^2 N_R}{z^4 \Gamma_1} \end{array}\right.$$

```
(* The following allows us to easily copy-paste the results into Matlab. *)
(*InputForm[PSC]*)
```

HGM computation

```
(* Evaluate the series expression for the outage probability. *)
OutProb[Γb_, Γ1_, K_, der_, NT_, NR_, c1_, τ_, trunc_, prec_] :=
  With[{
    NN = NR - NT + 1,
    (* If necessary, differentiate w.r.t. z, then plug in its value. *)
    zpart = D[Exp[-z] * z^n, {z, der}] /. z → K * NR * (NT - 1)},
   N[
    (* The part from G_n(z) in Eq.(67). *)
    Sum[Sum[Binomial[n, m] * Pochhammer[NN, m] / Pochhammer[NR + n - m, m] * c1^m *
        (* The part P_{o,m} from Eq.(60). *)
        Sum[Binomial[m, m1] * (-1)^m1 *
          (1 - Gamma[NN + m - m1, τ / Γ1] / Gamma[NN + m - m1]), {m1, 0, m}],
       (* The remaining parts from Eq.(67). *)
       {m, 0, n}] * zpart / n!, {n, 0, trunc}], prec]];

(* Here we produce the results shown in Fig. 1 using HGM. *)
RTKinv11 = 1.035478792935417;
c1 = 0.289132807530209;
With[{NT = 4, NR = 6, tau = dB[82 / 10], K0 = dB[-7]},
 Timing[data = Table[
     z0 = K0 * NR * (NT - 1);
     z1 = dB[K] * NR * (NT - 1);
     Gamma1 = 2 * dB[gb] / (dB[K] + 1) / RTKinv11;
     inits = Table[Derivative[d][Po][z0] ==
        OutProb[dB[gb], Gamma1, K0, d, NT, NR, c1, tau, 20, 20], {d, 0, 4}];
     ode = ApplyOreOperator[First[annOP], Po[z]] /.
       {N_R → NR, N → NR - NT + 1, c₁ → c1, τ → tau, Γ₁ → Gamma1};
     {gb, Po[z1]} /. First[NDSolve[Join[{ode == 0}, inits],
        Po, {z, z0, z1}, AccuracyGoal → 20]]
     , {K, 0, 21, 7}, {gb, 15, 25, 5 / 2}];]]
```

{4.020964, Null}

```
(* The plot shown in Fig. 1. *)
ListLogPlot[data, Joined → True, Ticks → {{15, 17.5, 20, 22.5, 25}, Automatic}]
```




```
(* Here are the numeric values. *)
TableForm[Map[Last, data, {2}]]
```

0.000804946	0.000151619	0.0000278514	5.04413×10^{-6}	9.06265×10^{-7}
0.0151968	0.00318437	0.000621553	0.000116505	0.0000213421
0.37649	0.128297	0.0335074	0.00743651	0.00150035
0.998383	0.937606	0.654186	0.294759	0.0926924

```
(* Compute the values of P_o for K = 0dB using the truncated series. *)
Timing[
 test1 = Table[OutProb[dB[gb], 2 * dB[gb] / (dB[0] + 1) / RTKinv11, dB[0], 0, 4, 6,
     c1, dB[82 / 10], 50, 10], {gb, 15, 25, 5 / 2}];]
{2.217000, Null}

(* Compare with HGM. *)
test1 / (Last /@ data[[1]])
{1., 1., 1., 1., 1.}

(* Compute the values of P_o for K = 7dB using the truncated series. *)
Timing[
 test2 = Table[OutProb[dB[gb], 2 * dB[gb] / (dB[7] + 1) / RTKinv11, dB[7], 0, 4, 6,
     c1, dB[82 / 10], 140, 10], {gb, 15, 25, 5 / 2}];]
{49.246000, Null}

(* Compare with HGM. *)
test2 / (Last /@ data[[2]])
{1., 1., 1., 1., 1.}

(* Compute the values of P_o for K = 14 dB using the truncated series. *)
(* Note that the duration of this computation is substantial. *)
Timing[
 test3 = Table[OutProb[dB[gb], 2 * dB[gb] / (dB[14] + 1) / RTKinv11, dB[14], 0, 4,
     6, c1, dB[82 / 10], 560, 10], {gb, 15, 25, 5 / 2}];]
{3893.454000, Null}

(* Compare with HGM. *)
test3 / (Last /@ data[[3]])
{1., 1., 1., 1., 1.}
```